
F08 – Least Squares and Eigenvalue Problems (LAPACK)

F08HCF (SSBEVD/DSBEVD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F08HCF (SSBEVD/DSBEVD) computes all the eigenvalues, and optionally all the eigenvectors, of a real
symmetric band matrix. If the eigenvectors are requested, then it uses a divide and conquer algorithm
to compute eigenvalues and eigenvectors. However, if only eigenvalues are required, then it uses the
Pal–Walker–Kahan variant of the QL or QR algorithm.

2 Specification

SUBROUTINE F08HCF(JOB, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK,
1 LWORK, IWORK, LIWORK, INFO)
ENTRY ssbevd(JOB, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK,
1 LWORK, IWORK, LIWORK, INFO)
INTEGER N, KD, LDAB, LDZ, LWORK, IWORK(∗), LIWORK, INFO
real AB(LDAB,∗), W(∗), Z(LDZ,∗), WORK(∗)
CHARACTER∗1 JOB, UPLO

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine computes all the eigenvalues, and optionally all the eigenvectors, of a real symmetric band
matrix A. In other words, it can compute the spectral factorization of A as

A = ZΛZT ,

where Λ is a diagonal matrix whose diagonal elements are the eigenvalues λi, and Z is the orthogonal
matrix whose columns are the eigenvectors zi. Thus

Azi = λizi for i = 1, 2, . . . , n.

4 References

[1] Golub G H and Van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd
Edition), Baltimore

5 Parameters

1: JOB — CHARACTER*1 Input

On entry: indicates whether eigenvectors are computed as follows:

if JOB = ’N’, then only eigenvalues are computed;
if JOB = ’V’, then eigenvalues and eigenvectors are computed.

Constraint: JOB = ’N’ or ’V’.

2: UPLO — CHARACTER*1 Input

On entry: indicates whether the upper or lower triangular part of A is stored as follows:

if UPLO = ’U’, then the upper triangular part of A is stored;
if UPLO = ’L’, then the lower triangular part of A is stored.

Constraint: UPLO = ’U’ or ’L’.

3: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: KD — INTEGER Input

On entry: k, the number of super-diagonals of the matrix A if UPLO = ’U’, or the number of
sub-diagonals if UPLO = ’L’.

Constraint: KD ≥ 0.

5: AB(LDAB,∗) — real array Input/Output

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the upper or the lower triangle of the n by n symmetric band matrix A, stored in the
first KD+1 rows of the array AB. More precisely, the jth column of A is stored in the jth column
of the array AB as follows:

if UPLO = ’U’, then AB(KD+1+i − j, j) = aij for max(1,j−KD) ≤ i ≤ j;
if UPLO = ’L’, then AB(1 + i − j, j) = aij for j ≤ i ≤ min(n, j+KD).

On exit: A is overwritten by the values generated during the reduction to tridiagonal form. If
UPLO = ’U’, the first superdiagonal and the diagonal of the tridiagonal matrix are returned in rows
KD and KD+1 of the array AB, respectively, and if UPLO = ’L’ then the diagonal and the first
subdiagonal of the tridiagonal matrix are returned in the first two rows of the array AB.

6: LDAB — INTEGER Input

On entry: the leading dimension of the array AB.

Constraint: LDAB ≥ KD+1.

7: W(∗) — real array Output

Note: the dimension of the array W must be at least max(1,N).

On exit: the eigenvalues of the matrix A in ascending order.

8: Z(LDZ, ∗) — real array Output

Note: the second dimension of the array Z must be at least max(1,N) if JOB = ’V’, and at least 1
if JOB = ’N’.

On exit: if JOB = ’V’, then this is overwritten by the orthogonal matrix Z which contains the
eigenvectors of A. The ith column of Z contains the eigenvector which corresponds to the eigenvalue
W(i).

If JOB = ’N’, then Z is not referenced.

9: LDZ — INTEGER Input

On entry: the leading dimension of the array Z as declared in the (sub)program from which F08HCF
(SSBEVD/DSBEVD) is called.

Constraints:

LDZ ≥ max(1,N) if JOB = ’V’;
LDZ ≥ 1 if JOB = ’N’.

10: WORK(∗) — real array Workspace

Note: the dimension of the array WORK must be at least LWORK.

On exit: if LWORK > 0, then WORK(1) contains the required minimal size of LWORK.

11: LWORK — INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08HCF
(SSBEVD/DSBEVD) is called.

Constraints:

if N ≤ 1, then LWORK ≥ 1;
if JOB = ’N’ and N > 1, then LWORK ≥ 2 × N;
if JOB = ’V’ and N > 1, then LWORK ≥ 3 × N2 + (4+2k) × N + 1 where k is the smallest
integer which satisfies 2k ≥ N.

12: IWORK(∗) — INTEGER array Workspace

Note: the dimension of the array IWORK must be at least LIWORK.

On exit: if LIWORK > 0, then IWORK(1) contains the required minimal size of LIWORK.

13: LIWORK — INTEGER Input

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08HCF
(SSBEVD/DSBEVD) is called.

Constraints:

if JOB = ’N’ or N ≤ 1, then LIWORK ≥ 1;
if JOB = ’V’ and N > 1, then LIWORK ≥ 5 × N + 2.

14: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

INFO > 0

If INFO = i, then the algorithm failed to converge; i indicates the number of elements of an
intermediate tridiagonal form which did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix A + E, where

‖E‖2 = O(ε)‖A‖2,

and ε is the machine precision.

8 Further Comments

The complex analogue of this routine is F08HQF (CHBEVD/ZHBEVD).

9 Example

To compute all the eigenvalues and eigenvectors of the symmetric band matrix A, where

A =




1.0 2.0 3.0 0.0 0.0
2.0 2.0 3.0 4.0 0.0
3.0 3.0 3.0 4.0 5.0
0.0 4.0 4.0 4.0 5.0
0.0 0.0 5.0 5.0 5.0




.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F08HCF Example Program Text.
* Mark 19 Release. NAG Copyright 1999.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, KDMAX, LDAB, LDZ
PARAMETER (NMAX=9,KDMAX=4,LDAB=KDMAX,LDZ=NMAX)
INTEGER LWORK, LIWORK
PARAMETER (LWORK=4*NMAX*NMAX,LIWORK=5*NMAX+2)

* .. Local Scalars ..
INTEGER I, IFAIL, INFO, J, KD, N
CHARACTER JOB, UPLO

* .. Local Arrays ..
real AB(LDAB,NMAX), W(NMAX), WORK(LWORK), Z(LDZ,NMAX)
INTEGER IWORK(LIWORK)

* .. External Subroutines ..
EXTERNAL ssbevd, X04CAF

* .. Intrinsic Functions ..
INTRINSIC MAX, MIN

* .. Executable Statements ..
WRITE (NOUT,*) ’F08HCF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, KD
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) UPLO
IF (UPLO.EQ.’U’) THEN

DO 20 I = 1, N
READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))

20 CONTINUE
ELSE IF (UPLO.EQ.’L’) THEN

DO 40 I = 1, N
READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)

40 CONTINUE
END IF

*
READ (NIN,*) JOB

*
* Calculate all the eigenvalues and eigenvectors of A
*

CALL ssbevd(JOB,UPLO,N,KD,AB,LDAB,W,Z,LDZ,WORK,LWORK,IWORK,
+ LIWORK,INFO)

*
WRITE (NOUT,*)
IF (INFO.GT.0) THEN

WRITE (NOUT,*) ’Failure to converge.’
ELSE

*
* Print eigenvalues and eigenvectors
*

WRITE (NOUT,*) ’Eigenvalues’
WRITE (NOUT,99999) (W(I),I=1,N)
WRITE (NOUT,*)
IFAIL = 0

*
CALL X04CAF(’General’,’ ’,N,N,Z,LDZ,’Eigenvectors’,IFAIL)

*
END IF

END IF
STOP

*
99999 FORMAT (3X,(8F8.4))

END

9.2 Program Data

F08HCF Example Program Data
5 2 :Values of N and KD
’L’ :Value of UPLO
1.0 2.0 3.0
2.0 2.0 3.0 4.0
3.0 3.0 3.0 4.0 5.0

4.0 4.0 4.0 5.0
5.0 5.0 5.0 :End of matrix A

’V’ :Value of JOB

9.3 Program Results

F08HCF Example Program Results

Eigenvalues
-3.2474 -2.6633 1.7511 4.1599 14.9997

Eigenvectors
1 2 3 4 5

1 0.0394 -0.6238 -0.5635 0.5165 0.1582
2 0.5721 0.2575 0.3896 0.5955 0.3161
3 -0.4372 0.5900 -0.4008 0.1470 0.5277
4 -0.4424 -0.4308 0.5581 -0.0470 0.5523
5 0.5332 -0.1039 -0.2421 -0.5956 0.5400

